2(mix)

WASHINGTON UNIVERSITY

Memorandum. No. 72/3 March, 1972

A COMPUTER PROGRAM FOR MAPPING SATELLITE-BORNE NARROW-BEAM ANTENNA FOOTPRINTS ON EARTH

Thomas W. Stagl
Jai P. Singh

WASHINGTON UNIVERSITY / ST. LOUIS / MISSOURI 63130

29

PROGRAM ON APPLICATION OF COMMUNICATIONS SATELLITES
TO EDUCATIONAL DEVELOPMENT

WASHINGTON UNIVERSITY

Memorandum No. 72/3 March, 1972

A COMPUTER PROGRAM FOR MAPPING SATELLITE-BORNE NARROW-BEAM ANTENNA FOOTPRINTS ON EARTH

Thomas W. Stagl
Jai P. Singh

This research is supported by the National Aeronautics and Space Administration under Grant No. Y/NGL-26-008-054. This memorandum is primarily for internal distribution and does not necessarily represent the views of either the research team as a whole or the National Aeronautics and Space Administration.

Table of Contents

A COMPUTER PROGRAM FOR MAPPING SATELLITE-BORNE NARROW-BEAM ANTENNA FOOTPRINTS ON EARTH

I. INTRODUCTION

A computer program has been developed that computes the locus of intersection of a quadric cone and a sphere. The outputs of the program are a list of the longitude and latitude coordinates of the locus of intersection and a plot of the locus. It was written primarily to define the area of the earth covered by a narrow-beam antenna carried on a synchronous satellite in circular, near equatorial orbits.

The program is basically an implementation of a report by S. L. Zolnay[1] with some modifications added. The main modifications are the incorporation of an elliptical cross section antenna beam and computation of the beam vertex angles and corresponding locus of intersection for any signal level between 0.1 and 10 db down from the beam center. Any number of signals up to 10 may be input for each data set. The output will be plotted on the same graph.

The program was written for use with a Cal Comp 570/563 off-line plotting system and uses the standard Cal Comp subroutines supplied with the system. The plot is drawn using linear longitude and latitude scales and non-linear scales such as Mercator scales cannot be used.

It should be noted that this program assumes the earth to be spherical rather than oblate as it actually is. However, by assuming a spherical earth, one introduces maximum error distances of about 11.5 nautical miles. For many purposes, this error can be ignored.

II. MATHEMATICAL ANALYSIS

Figure 1 shows the intersection geometry. The satellite is located at point SAT in a circular, near-equatorial orbit which is a distance DIST from center of the earth. Point P is the point of intersection of satellite antenna boresight (or beam center) and the earth. AL is the arc from the point on earth directly below the satellite (the subsatellite point) to P and is given by:

$$AL = \mathrm{Cos}^{-1}\,[\mathrm{Cos}(LONCTR)\mathrm{Cos}(LACTR + DELT)] \tag{1}$$

Where LATCTR is latitude of P, LONCTR is the longitude of P relative to the subsatellite point, and DELT is the instantaneous declination angle formed by a vector from satellite to the center of the earth and the equatorial plane. At DIST = ~~30,220~~ 22,300 statute miles (geostationary equatorial orbit when DELT = 0), the maximum arc, AL, permissible for the point to be seen by the satellite is 81.3°. When AL is computed to be larger than 81.3°, the point is over the horizon seen by the satellite and any attempt at finding the locus of intersection would produce meaningless results. Therefore, the computation stops at this point.

The next step is to calculate the vector, $\underline{RS}$, extending from the satellite to point P. A right-handed coordinate system is constructed with the origin at SAT, the positive Y axis extending through the earth center and the Z axis in the plane defined by SAT and the north and south poles. The vector $\underline{RS}$ is given in terms of coordinates along these axes:

$$\begin{aligned}\underline{RS} = {} & RE\cos(LATCTR)\sin(LONCTR)\,\underline{i} + \\ & [DIST\cos(DELT) - RE\cos(LATCTR)\cos(LONCTR)]\,\underline{j} + \\ & [DIST\sin(DELT) + RE\sin(LATCTR)]\,\underline{k}\end{aligned} \tag{2}$$

Where RE is radius of earth, and $\underline{i}$, $\underline{j}$ and $\underline{k}$ are unit vectors in the positive x, y, and z directions, respectively.

Pitch and roll angles are then defined as in Figure 2. These are given by:



FIGURE I GEOMETRY OF INTERSECTION



FIGURE 2 GRAPHIC DESCRIPTION OF PITCH AND ROLL ANGLES

$$\text{PITCH} = \text{Sin}^{-1} \frac{\underline{RS}_{\underline{i}}}{\sqrt{RS_{\underline{i}}^{2} + RS_{\underline{j}}^{2}}} \tag{3}$$

$$\text{ROLL} = \text{Tan}^{-1} \frac{RS_{\underline{k}}}{\sqrt{RS_{i}^{2} + RS_{\underline{j}}^{2}}} \tag{4}$$

A new coordinate system is defined by rotating the old system about its z-axis by an angle equal to PITCH, and then about the resultant x-axis by an angle equal to ROLL.

A vector U is then defined as lying in a plane perpendicular to vector RS. The origin of vector U is at the point of intersection of RS and that plane, see Figure 3.

The equation defining U is:

$$\underline{U} = [\cos(\text{BETA})\ \underline{i}' + \sin(\text{BETA})\ \underline{k}']\ [A^{-2}\cos^{2}(\text{BETA}) + B^{-2}\sin^{2}(\text{BETA})]^{-\frac{1}{2}} \tag{5}$$

where $\underline{i}'$ and $\underline{k}'$ are unit vectors in the new coordinate system. This equation is a parametric equation of an ellipse with BETA as the parameter and A and B corresponding to the semi-major and semi-minor axes, respectively.

A ray on the surface of the beam can now be generated by a vectorial addition of RS and U. By incrementing BETA from 0° to 360° the entire outer surface of the beam can be generated.

Let a vector Mn be a surface generator vector. The equation defining Mn in the new coordinate system is:

$$\begin{aligned}\underline{Mn} = {} & [|\underline{RS}|\cos(\text{ROLL})\sin(\text{PITCH}) + |\underline{U}|\cos(\text{BETA})\cos(\text{PITCH}) \\ & - |\underline{U}|\sin(\text{BETA})\sin(\text{ROLL})\sin(\text{PITCH})]\ \underline{i}' \\ & + [|\underline{RS}|\cos(\text{ROLL})\cos(\text{PITCH}) + |\underline{U}|\cos(\text{BETA})\sin(\text{PITCH}) \\ & - |\underline{U}|\sin(\text{BETA})\sin(\text{ROLL})\cos(\text{PITCH})]\ \underline{j}' \\ & + [|\underline{RS}|\sin(\text{ROLL}) + |\underline{U}|\sin(\text{BETA})\cos(\text{ROLL})]\ \underline{k}'\end{aligned} \tag{6}$$



FIGURE 3 GRAGHIC DESCRIPTION OF BEAM CROSS SECTION

In the old coordinate system:

$$\underline{Mn} = [Mn_{\underline{i}'}]\ \underline{i} + [Mn_{\underline{j}'} \cos(DELT) + Mn_{\underline{k}'} \sin(DELT)]\ \underline{j}$$
$$+ [-Mn_{\underline{j}'} \sin(DELT) + Mn_{\underline{k}'} \cos(DELT)]\ \underline{k} \qquad (7)$$

Pitch and roll angles are defined for each $\underline{Mn}$ as follows:

$$PITCH\ N = Sin^{-1} \frac{Mn_{\underline{i}'}}{\sqrt{Mn^2_{\underline{i}'} + Mn^2_{\underline{j}'}}} \qquad (8)$$

$$ROLL\ N = Tan^{-1} \frac{Mn_{\underline{k}'}}{\sqrt{Mn_{\underline{i}'} + Mn_{\underline{j}'}}} \qquad (9)$$

The earth radius vector to the point of intersection of $\underline{Mn}$ is:

$$\underline{RE} = [Mn \cos(ROLLN) \sin(PITCHN)]\ \underline{I}$$
$$+ [DIST \cos(DELT) - Mn \cos(ROLLN) \cos(PITCHN)]\ \underline{J}$$
$$+ [Mn \sin(ROLLN) - DIST \sin(DELT)]\ \underline{K} \qquad (10)$$

Where Mn in the length of $\underline{Mn}$ and $\underline{I}$, $\underline{J}$, and $\underline{K}$ are unit vectors in an earth centered coordinate system whose positive z axis extends through the north pole and positive x axis intersects the 0° meridian.

The latitude and relative longitude coordinates of the point of intersection of $\underline{Mn}$ are found by:

$$LAT = Tan^{-1} \frac{RE_{\underline{K}}}{\sqrt{RE^2_{\underline{I}} + RE^2_{\underline{J}}}} \qquad (11)$$

$$LON = Sin^{-1} \frac{RE_{\underline{I}}}{\sqrt{RE^2_{\underline{I}} + RE^2_{\underline{J}}}} \qquad (12)$$

The actual longitude is found by adding the above longitude to the longitude of the subsatellite point.

The relative beamwidth for any signal level is found using a beamwidth conversion chart. The particular chart used in this program is taken from Microwave Engineers Handbook[2], (see Figure 4). The actual conversion in the program is done by using linear interpolation between the appropriate two consecutive points from the following set:

(.1, .18) (.2, .26) (.5, .4) (1., .56) (1.5, .7) (3., 1.) (5., 1.27) (10., 1.7)

These points were chosen such that the graph segment between any two consecutive points is approximately linear.

The above discussion assumes that the entire antenna beam intersects the earth. The case when the boresight location is near enough to the horizon that a portion of the beam passes the earth is considered next.

Define an angle, Bn formed by the vector $\underline{\text{Mn}}$ and the vector from the satellite to earth center. The angle at which $\underline{\text{Mn}}$ is tangent to the earth is given, from the law of sines, as:

$$B_{max} = \text{Sin}^{-1} \frac{RE}{DIST} \tag{13}$$

When the angle B_n is larger than B_{max} the earth radius vector $\underline{RE}$ must be computed to the point of tangency, i.e. the horizon seen by the satellite. The vector $\underline{RE}$ is then defined by:

$$\begin{aligned} \underline{RE} = {} & RE\,[\cos(B_{max})\,\sin(TAUN)]\,\underline{I} \\ & - RE\,[\cos(DELT)\,\sin(B_{max}) \\ & + \sin(DELT)\,\cos(B_{max})\,\cos(TAUN)]\,\underline{J} \\ & - RE\,[\sin(DELT)\,\sin(B_{max}) \\ & - \cos(DELT)\,\cos(B_{max})\,\cos(TAUN)]\,\underline{K} \end{aligned} \tag{14}$$

When TAUN is the tilt angle defined by:

$$TAUN = \text{Cos}^{-1} \frac{Mn_{\underline{k}}}{\sqrt{Mn_{\underline{i}}^2 + Mn_{\underline{k}}^2}} \tag{15}$$



FIGURE 4 BEAMWIDTH CONVERSION

and RE is the radius of the earth. The latitude and relative longitude coordinate are found as in equations (11) and (12).

III. USAGE

The inputs to the program are:

LONSS - longitude of point on earth directly below satellite (°E)

LONCT - longitude of boresight intersection (°E)

LATCT - latitude of boresight intersection (°N)

INCR - increments of angular parameter BETA, i.e. number of points plotted = 360°/INCR

DELT - instantaneous declination of satellite - earth center vector from plane of equator (°below equatorial plane)

THETA - orientation angle of elliptical beam about beam center (measured positive counter-clockwise)

BW1 - beam vertex angle in plane defined by beam vertex and major axis of beam cross section at point 3 db below beam center

BW2 - beam vertex angle in plane defined by beam vertex and minor axis of beam cross section at point 3 db below beam center

L - number of signal levels to be plotted

DB(I) - signal levels to be plotted (measured in db below beam center)

It should be noted that all inputs except L and DB(I) are in degrees rather than radians. Longitude inputs can be expressed as degrees east of 0° or negative degrees west of 0°. The output will agree with the input.

Each input data set will consist of three cards. The format for the data is:

Card #1

column	
1 - 9	LONSS
10 - 18	LONCT
19 - 27	LATCT
28 - 36	INCR
37 - 45	DELT

Card #2

1 - 9	THETA
10 - 18	BW1
19 - 27	BW2
28 - 29	L

Card #3

1 - 7	
8 - 14	DB(I)
15 - 21	

Every input, with the exception of L, must contain a decimal point. The input L must not contain a decimal point and must be right justified in columns 28-29, i.e. single digit input must be in column 29. The inputs DB(I) must be input in increasing order with the signal level closest to beam center listed first and the level farthest from beam center listed last.

The output of the program consists of two parts, the printout and the plot. The printout contains, for each set of input data, a list of the following parameters:

SUB SAT LONG - longitude of point directly below satellite

BOR SGHT LONG - longitude of boresight intersection

BOR SGHT LAT - latitude of boresight intersection

DECLINATION - instantaneous declination of satellite-earth center vector from equatorial plane

MIN BMWDTH - beamwidth along minor axis of beam cross section at -3 db (half-power) level

MAX BMWDTH - beamwidth along major axis of beam cross section at -3 db (half-power) level

ORIENTATION - orientation angle of beam about beam center

ELEVATION - angle formed by vector <u>RS</u> and a plane tangent to earth at boresight intersection

The remainder of the printout gives the maximum and minimum beamwidth and a listing of the coordinates of the locus of intersection for each signal level.

Each set of input data is given a data set number. This number appears on the printout and the plot for each data set. This facilitates matching of plots with printout when more than one data set is run.

A modification of this program has recently been developed to plot the coverage of a multi-beam satellite. This new program computes the locus of intersection of a number of sets of input data as does the original. However, the modification plots all of the intersection loci on one set of axes so that uncovered areas and overlapped areas are immediately obvious.

The major changes to the original program are:

1. The plotter tape is opened and the axes are drawn before the main calculation begins.
2. The axes are scaled once at the beginning of the program rather than being scaled for each data set.
3. The computed coordinates are checked to see that they do not extend beyond the limits of the axes.
4. The plot origin is not reset for each data set.

IV. TYPICAL EXAMPLES

Figures 5-9 present the area coverage plots for input values shown in Table 1. For each case, 3, 5 and 10 dB level contours are plotted. The inputs represent a wide variety of cases--satellites positioned in circular, equatorial and geosynchronous orbits; satellites positioned in slightly inclined stationary orbits; and area coverage at small inclination angles.

Figure 10 shows the coverage provided by a 4-beam satellite positioned at 120°W longitude. All plots use the same set of axes. The input values for various beams are given in Table 2. All four beams are designed to provide complete coverage to the United States--beam 1 covers Hawaii, beam 2 covers Alaska, and beams 3 and 4 provide coverage to the other 48 states.

B

Table 1. Input Values for Coverage Plots shown in Figures 5-9

Input	Case I (Figure 5)	Case II (Figure 6)	Case III (Figure 7)	Case IV (Figure 8)	Case V (Figure 9)
Subsatellite Longitude	-115.00	-115.00	-115.00	-115.00	-115.00
Boresight Longitude	-157.00	-157.00	- 74.00	- 74.00	- 74.00
Boresight Latitude	21.30	21.30	40.75	40.75	40.75
Declination (in degrees)	0.00	0.00	0.00	1.00	0.00
Minor-axis Beamwidth (in degrees)	1.50	0.75	1.00	1.00	1.00
Major-axis Beamwidth (in degrees)	1.50	1.50	1.00	1.00	0.50
Orientation (in degrees)	0.00	- 25.00	0.00	0.00	40.00


DATA SET NO. 1
LATITUDE (DEGREES)
49.00
45.00
41.00
37.00
33.00
29.00
25.00
21.00
17.00
13.00
10 dB
5 dB
3 dB
Beam Center
-176.00
-172.00
-168.00
-164.00
-160.00
-156.00
-152.00
-148.00
-144.00
LONGITUDE (DEGREES)

Figure 5


DATA SET NO. 2
10 dB
5 dB
3 dB
LATITUDE (DEGREES)
46.00
42.00
38.00
34.00
30.00
26.00
22.00
18.00
14.00
-176.00
-172.00
-168.00
-164.00
-160.00
-156.00
-152.00
-148.00
-144.00
LONGITUDE (DEGREES)

Figure 6


DATA SET NO. 3
LATITUDE (DEGREES)
65.00
61.00
57.00
53.00
49.00
45.00
41.00
37.00
33.00
LONGITUDE (DEGREES)
-88.00
-84.00
-80.00
-76.00
-72.00
-68.00
-64.00
-60.00
-56.00
10 dB
5 dB
3 dB

Figure 7


DATA SET NO. 4
LATITUDE (DEGREES)
65.00
61.00
57.00
53.00
49.00
45.00
41.00
37.00
33.00
10 dB
5 dB
3 dB
-85.00
-81.00
-77.00
-73.00
-69.00
-65.00
-61.00
-57.00
-53.00
LONGITUDE (DEGREES)

Figure 8


DATA SET NO. 5
10 dB
5 dB
3 dB
Beam Center
LATITUDE (DEGREES)
48.00
46.00
44.00
42.00
40.00
38.00
36.00
34.00
32.00
-82.00
-80.00
-78.00
-76.00
-74.00
-72.00
-70.00
-68.00
-66.00
LONGITUDE (DEGREES)

Figure 9

Table 2. Input Values for 4-Beam Coverage Plot Shown in Figure 10

Input	Hawaii	Alaska	Beam I for western and mountain states	Beam II for eastern and central states
Subsatellite Longitude	-120.00	-120.00	-120.00	-120.00
Boresight Longitude	-158.00	-145.00	-112.00	- 88.00
Boresight Latitude	21.00	63.00	37.00	36.00
Satellite Declination (in degrees)	0.00	0.00	0.00	0.00
Minor-axis Beamwidth (in degrees)	1.50	1.50	3.30	3.30
Major-axis Beamwidth (in degrees)	2.00	2.30	3.40	3.70
Orientation (in degrees)	135.00	110.00	0.00	0.00



Figure 10. Four Beam Coverage of U.S. (see Table 2)

V. REFERENCES

1. S. L. Zolnay, "Earth Coverage ('Footprint') of a Satellite-Borne Antenna", Technical Note 1971-7, Lincoln Laboratory, Massachusetts Institute of Technology, Lexington, Massachusetts (February 5, 1971).

2. Theodore S. Saad, "Antennas" in Microwave Engineers Handbook (Vol. 2), Dedmam, Massachusetts: Artech House, Inc., p. 24.

3. W. Solfrey, "Earth Coverage Patterns with High Gain Antenna on Stationary Satellites", Memorandum RM-4894-NASA, The Rand Corporation, Santa Monica, California (1967).

APPENDIX A

PROGRAM LISTING

```
0001            IMPLICIT REAL*8(A-Z)
0002            REAL*8 DSIN,DCOS,DSQRT,DARSIN,DARCOS,DATAN
0003            INTEGER*4 NSTEPS,ICT,I,N,NPLT,IBUF(1000),NP1,NP2,NDB,LDB
0004            INTEGER*4 J,J1,L,K
0005            INTEGER*4 FMT(4)/'(',' ','F7.3',')'/
0006            INTEGER*4 DIGIT(10)/'1','2','3','4','5','6','7','8','9','10'/
0007            REAL*4 LON(722),LAT(722),N1,XCT,YCT
0008            DIMENSION DB(10),DBS(8),RBWS(8)
0009            DATA DBS/.1,.2,.5,1.,1.5,3.,5.,10./
0010            DATA RBWS/.18,.26,.4,.56,.7,1.,1.27,1.7/
0011            DATA PI/3.14159265358979300/,RE/3.96D3/,DIST/2.626D4/
0012            N=0
0013            NPLT=0
0014            CONVTR=PI/180.
0015            BM=DARSIN(RE/DIST)
0016            CHK=81.3*CONVTR
0017          1 READ(5,500,END=1000)DLONSS,DLONCT,DLATCT,DINCR,DDELT
0018        500 FORMAT(5F9.3)
0019            READ(5,501)THETA1,BW1A,BW2A,L
0020        501 FORMAT(3F9.3,I2)
0021            FMT(2)=DIGIT(L)
0022            READ(5,FMT)(DB(I),I=1,L)
0023            NDB=0
0024            N=N+1
0025            WRITE(6,600)N,DLONSS,DLONCT,DLATCT,DDELT,BW2A,BW1A,THETA1
0026        600 FORMAT(//'1DATA SET NO. ',I2/4X,'SUB SAT LONG   =',F9.3/
               1  4X,'BOR SGHT LONG  =',F9.3/4X,'BOR SGHT LAT   =',F9.3/
               2  4X,'DECLINATION    =',F9.3/4X,'MIN BMWDTH',5X,'=',F9.3/
               3  4X,'MAX BMWDTH',5X,'=',F9.3/4X,'ORIENTATION    =',F9.3)
          C  CONVERT FROM DEGREES TO RADIANS
0027            INCR=DINCR*CONVTR
0028            LONCTR=(DLONCT-DLONSS)*CONVTR
0029            LATCTR=DLATCT*CONVTR
0030            DELT=DDELT*CONVTR
0031            BW1=BW1A*CONVTR/2.0
0032            BW2=BW2A*CONVTR/2.0
0033            THETA=THETA1*CONVTR
          C  CHECK FOR BORESIGHT LOCATION IN RANGE OF SATELLITE
0034            AL=DARCOS(DCOS(LONCTR)*DCOS(LATCTR+DELT))
0035            IF(AL.LE.CHK)GO TO 21
0036         20 WRITE(6,400)
0037        400 FORMAT(1H0,'BORESIGHT LOCATION IS NOT IN RANGE OF SATELLITE.'/)
0038            GO TO 1
          C  COMPUTE VECTOR FROM SATELLITE TO BORESIGHT LOCATION
0039         21 RSX=RE*DCOS(LATCTR)*DSIN(LONCTR)
0040            RSY=DIST*DCOS(DELT)-RE*DCOS(LATCTR)*DCOS(LONCTR)
0041            RSZ=DIST*DSIN(DELT)+RE*DSIN(LATCTR)
0042            RSM=DSQRT(RSX**2+RSY**2+RSZ**2)
```

```
0043          GAM=DARSIN(RE*DSIN(AL)/RSM)
0044          EL=((PI/2.0)-(AL+GAM))/CONVTR
0045          WRITE(6,300) EL
0046      300 FORMAT(4X,'ELEVATION',6X,'=',F9.3)
0047          DENOM=DSQRT(RSX**2+RSY**2)
0048          PITCH=DARSIN(RSX/DENOM)
0049          ROLL=DATAN(RSZ/DENOM)
0050          CP=DCOS(PITCH)
0051          CR=DCOS(ROLL)
0052          SP=DSIN(PITCH)
0053          SR=DSIN(ROLL)
0054          NPLT=NPLT+1
0055          IF(NPLT.GT.1)GO TO 15
        C  FIRST PLOT, OPEN PLOTTAPE AND SET ORIGIN
0056          CALL PLOTS(IBUF,1000)
0057          CALL PLOT(1.0,1.0,23)
0058          GO TO 12
        C  NOT FIRST PLOT, SET ORIGIN
0059       15 CALL PLOT(1.0,1.0,-3)
0060       12 DO 31 K=1,L
0061          NDB=NDB+1
0062          LDB=L-K+1
        C  INTERPOLATE FOR RELATIVE BEAMWIDTH
0063          DO 8 J=1,8
0064          IF(DB(LDB)-DBS(J))6,7,8
0065        7 RBW=RBWS(J)
0066          GO TO 9
0067        8 CONTINUE
0068        6 J1=J-1
0069          RBW=(DB(LDB)-DBS(J1))*(RBWS(J)-RBWS(J1))/(DBS(J)-DBS(J1))
             1        +RBWS(J1)
0070        9 A=RSM*DTAN(BW1*RBW)
0071          B=RSM*DTAN(BW2*RBW)
0072          ALPH1=BW1A*RBW
0073          ALPH2=BW2A*RBW
0074          NSTEPS=(360.0/DINCR)+1
0075          BMWDTH=DALPHA*RBW
0076          DO 10 ICT=1,NSTEPS
0077          BETAN=(ICT-1)*INCR
0078          ANG=BETAN+THETA
0079          UM=1.0/DSQRT((DCOS(BETAN)/A)**2+(DSIN(BETAN)/B)**2)
        C  COMPUTE N-TH M VECTOR FROM SATELLITE TO LOCUS ON EARTH
0080          MNPX=RSM*CR*SP+UM*DCOS(ANG)*CP-UM*DSIN(ANG)*SR*SP
0081          MNPY=RSM*CR*CP+UM*DCOS(ANG)*SP-UM*DSIN(ANG)*SR*CP
0082          MNPZ=RSM*SR+UM*DSIN(ANG)*CR
0083          MNX=MNPX
0084          MNY=MNPY*DCOS(DELT)+MNPZ*DSIN(DELT)
0085          MNZ=MNPY*(-DSIN(DELT))+MNPZ*DCOS(DELT)
```

```
0086            BN=DARCOS(MNY/DSQRT(MNX**2+MNY**2+MNZ**2))
0087            IF(BN.GT.BM)GO TO 30
0088            CN=PI-DARSIN(DSIN(BN)*DIST/RE)
0089            DN=PI-(BN+CN)
0090            MNL=DSQRT(RE**2+DIST**2-2.*RE*DIST*DCOS(DN))
0091            DENOM=DSQRT(MNPX**2+MNPY**2)
0092            PITCHN=DARSIN(MNPX/DENOM)
0093            ROLLN=DATAN(MNPZ/DENOM)
          C  COMPUTE N-TH VECTOR FROM CENTER OF EARTH TO LOCUS
0094            REI=MNL*DCOS(ROLLN)*DSIN(PITCHN)
0095            REJ=DIST*DCOS(DELT)-MNL*DCOS(ROLLN)*DCOS(PITCHN)
0096            REK=MNL*DSIN(ROLLN)-DIST*DSIN(DELT)
0097            GO TO 40
          C  IF N-TH M VECTOR DOES NOT INTERSECT EARTH COMPUTE VECTOR
          C   FROM CENTER OF EARTH TO HORIZON SEEN BY SATELLITE
0098         30 TAUN=DARCOS(MNZ/DSQRT(MNX**2+MNZ**2))
0099            IF(MNX.LT.0.0)TAUN=-TAUN
0100            REI=RE*DCOS(BM)*DSIN(TAUN)
0101            REJ=RE*(-DCOS(DELT)*DSIN(BM)-DSIN(DELT)*DCOS(BM)*DCOS(TAUN))
0102            REK=RE*(-DSIN(DELT)*DSIN(BM)+DCOS(DELT)*DCOS(BM)*DCOS(TAUN))
0103         40 DEN=DSQRT(REI**2+REJ**2)
          C  COMPUTE LONGITUDE AND LATITUDE COORDINATES
0104            LAT(ICT)=SNGL(DATAN(REK/DEN)/CONVTR)
0105         10 LON(ICT)=SNGL(DARSIN(REI/DEN)/CONVTR+DLONSS)
0106            WRITE(6,100)DB(LDB),ALPH1,ALPH2
0107        100 FORMAT(1H-,3X,'AT ',F4.1,' DB LEVEL:'/4X,'MAX BMWDTH= ',F5.2/
               1        4X,'MIN BMWDTH= ',F5.2//
               2        5X,'LONGITUDE',7X,'LATITUDE'/5X,'(DEGREES)',6X,'(DEGREES)'
               3        /)
0108            DO 80 I=1,NSTEPS
0109         80 WRITE(6,200) LON(I),LAT(I)
0110        200 FORMAT(1H ,4X,F9.3,6X,F9.3)
0111            IF(NDB.GT.1)GO TO 93
          C  IF FIRST DB LEVEL PLOT AXES AND TITLES
0112            CALL SCALE(LON,9.0,NSTEPS,1)
0113            CALL SCALE(LAT,9.0,NSTEPS,1)
0114            NP1=NSTEPS+1
0115            NP2=NSTEPS+2
0116            IF(LON(NP2)-LAT(NP2))90,91,92
0117         90 LON(NP2)=LAT(NP2)
0118            GO TO 91
0119         92 LAT(NP2)=LON(NP2)
0120         91 CALL AXIS(0.0,0.0,'LONGITUDE (DEGREES)',-19, 9.0,0.0,LON(NP1),
               1        LON(NP2))
0121            CALL AXIS(0.0,0.0,'LATITUDE (DEGREES)',18, 9.0,90.0,LAT(NP1),
               2        LAT(NP2))
0122            SCFCT=LON(NP2)
0123            EVLN=LON(NP1)
```

```
 0124              FVLT=LAT(NP1)
 0125              XCT=(SNGL(DLONCT)-LON(NP1))/LON(NP2)
 0126              YCT=(SNGL(DLATCT)-LAT(NP1))/LAT(NP2)
 0127              CALL SYMBOL(XCT,YCT,0.21,3,0.0,-1)
 0128              CALL SYMBOL(1.0,8.0,0.21,'DATA SET NO. ',0.0,13)
 0129              N1=FLOAT(N)
 0130              CALL NUMBER(999.0,999.0,0.21,N1,0.0,-1)
 0131           93 LON(NP2)=SCFCT
 0132              LAT(NP2)=SCFCT
 0133              LON(NP1)=FVLN
 0134              LAT(NP1)=FVLT
             C  PLOT COORDINATES OF LOCUS
 0135              CALL LINE(LON,LAT,NSTEPS,1,0,0)
 0136           31 CONTINUE
 0137              CALL PLOT(14.0,-1.0,23)
 0138              GO TO 1
 0139         1000 IF(NPLT.LE.0) GO TO 1001
             C  CLOSE PLOTTAPE
 0140              CALL PLOT(0.0,0.0,999)
 0141         1001 WRITE(6,700) NPLT
 0142          700 FORMAT('1NUMBER OF PLOTS PRODUCED =',I3)
 0143              STOP
 0144              END
```